
1 

Paper 163-30 

A Macro for Computing Point Estimates and Confidence Intervals for 
Mahalanobis Distance 

Kristine Y. Hogarty  
John M. Ferron 

Melinda R. Hess 
 Jeffrey D. Kromrey 

University of South Florida, Tampa, FL 
 

ABSTRACT 
This paper discusses a SAS® macro that provides three approaches to statistical inferences about Mahalanobis 
distance.  Mahalanobis distance is useful as a multivariate effect size, being an extension of the standardized mean 
difference (i.e., Cohen’s d).  This program calculates three point estimates of D2 (a sample estimate, a jackknife 
estimate, and an adjusted estimate advanced by Rao, 1973).  Further, the program computes a series of confidence 
bands around each estimated value of D2.  These confidence intervals are calculated employing two estimation 
procedures (an interval inversion approach and a common bootstrapping method).  The paper presents a 
demonstration of the SAS/IML® code employed in a macro entitled D2BAND, and provides a table with the resulting 
output.  

INTRODUCTION 
As is true of scholars in many fields, increasing energy has been directed toward encouraging educational 
researchers to provide comprehensive and accurate analyses of empirical data and thorough, informative reporting of 
results.  Two aspects of statistical analysis that have gained attention over recent years are the use of effect sizes 
and the application of confidence intervals. The fifth edition of the American Psychological Association’s style manual 
for publication (APA, 2001) cites the failure to report effect sizes, as well as other research issues, as defects in 
reporting research.  However, Thompson (1998) has noted that the ‘encouragement’ of the APA has not seemed to 
induce sufficient leverage for researchers to consistently report these informative statistics.  As an incentive for 
researchers to report and interpret effect sizes, some professional journals require authors to supply this information 
before consideration for publication.   Unfortunately, recent research has indicated that while many journals 
technically require this information, few are enforcing their own mandates (McMillan, Snyder & Lewis, 2002).   
 
The report by Wilkinson and the APA Task Force on Statistical Inference (1999) not only addresses the need for 
effect size reporting but also stresses the obligation of researchers to provide estimates of confidence intervals for all 
principal outcomes, including, but not limited to, effect size information. In recent years, research methodologists have 
investigated the use of confidence intervals (in lieu of traditional point estimates) for the estimation of various 
parameters of interest (see, for example, Nix & Barnette, 1998; Grissom & Kim, 2001).  The use of confidence 
intervals for effect sizes has also been a topic of empirical study (Hess & Kromrey, 2003; Hess & Kromrey 2002; 
Cumming & Finch, 2001; Fidler & Thompson, 2001; Steiger & Fouladi, 1992). Such investigations, however, have 
focused on univariate effect sizes.  
 
In a recent Monte Carlo Study we investigated the accuracy and precision of confidence bands for two-group 
multivariate effect sizes under a variety of data conditions, including nonnormality of population distributions and 
covariance matrix heterogeneity.  Our findings led to recommendations regarding promising methods for calculating 
point estimate of multivariate effect size as well as useful approaches to constructing confidence bands.  Therefore, 
the purpose of this paper is to provide a mechanism for other researchers who wish to estimate multivariate estimates 
of effect size and their corresponding confidence intervals. 
 
MULTIVARIATE EFFECT SIZES 
A commonly used univariate effect size for the two-group case is the standardized mean difference, Cohen’s d.  
Cohen’s d is defined as the difference in means between groups divided by the pooled standard deviation and is 
given by: 

where 1 2, and pX X s  are the sample means of group 1 and 2, and the pooled standard deviation, 

respectively. 

A multivariate extension of the standardized mean difference is the Mahalanobis Distance: 

where iX

%
 is the sample mean vector of group i and Σ̂ is the pooled covariance matrix. 

 
As a point estimate, the statistic D2 is known to evidence small sample bias (Krzanowski & Marriott, 1995). To reduce 
such bias, an ‘adjusted’ value of D2 was suggested by Rao (1973): 

where n1 and n2 are the sample sizes of the two groups and p is the number of variables in the covariance 

matrix. 
 
A second approach to the reduction of bias in D2 is the Jackknife estimator (Rao & Dorvlo, 1985): 

where 2D is the mean value of D2 obtained from successively removing each observation (i.e., the mean of 

the leave-one-out values of D2).  The leave-one-out values of D2, the 2
ijD , can be more explicitly defined by  

where ijX

%
 and ijS are the sample mean vector and covariance matrix of group i after observation j has been 

removed from group i, and mX
%

is the sample mean vector for the other group. 
 
With input from the researcher, the MACRO D2BAND currently computes all three of the aforementioned measures of 
multivariate effect size. 
 
CONFIDENCE INTERVALS FOR MULTIVARIATE EFFECT SIZES 
Several methods for constructing confidence bands around the multivariate point estimates of effect size have been 
suggested. An elegant approach for constructing confidence intervals around D2 is the interval inversion approach 
(Reiser, 2001). The computational details of such an approach were lucidly described by Steiger and Fouladi (1992, 
1997), although this approach to confidence interval estimation was detailed earlier (see, for example, Venables, 
1975; Serlin & Lapsley, 1985).  Interval inversion has been shown to provide accurate confidence bands around 
univariate effect sizes in a variety of conditions (Kromrey & Hess, 2001; Hess & Kromrey, 2002). This approach seeks 
the population values of D2 for which the sample value would be expected (for example) 2.5% of the time and 97.5% 
of the time. Such values provide the endpoints of a 95% confidence interval. In its execution, however, the approach 
is based on a function of D2: 

which is distributed as F with df = p, n1 + n2 – p – 1 and the noncentrality parameter λ  
 

where 21 2

 and 2δ is the population value of the Mahalanobis Distance. 

 
Thus, the interval inversion approach seeks the values of noncentrality that correspond to the target percentiles of the 
appropriate F distribution. 
 
BOOTSTRAP METHODS 
Bootstrapping is commonly recognized as an efficient method for providing estimates for, among other things, 
confidence intervals (Efron & Gong, 1983;  Efron & Tibshirani, 1986; Stine, 1990).  In this procedure, samples of 
observations are drawn with replacement from the observed data, and the statistic of interest (e.g., D2, Rao’s adjusted 
D2) is computed on the resampled observations.  Over the years, a variety of algorithms have been proposed to 
construct bootstrap intervals.  Carpenter and Bithell (2000) investigated eight different bootstrap algorithms using data 
from a medical study to address this issue.  While their study used a different statistic of interest, the methods they 

used appear to be generalizable to other statistics, such as multivariate effect sizes.  For our initial investigation, four 
methods were identified as promising for further research.  Here we present the bootstrapping method that provided 
the highest level of accuracy and precision. 
 
This basic method of bootstrapping (called the naïve bootstrap or the percentile method) consists of repeatedly 
drawing samples of size n with replacement (e.g., 1000 times or 5000 times) from a single sample of n observations.  
Each bootstrap sample provides an estimate of the effect size, and the set of estimates provides an empirical 
sampling distribution for the effect size. Percentile points in this empirical sampling distribution (e.g., the 2.5th 
percentile and the 97.5th percentile) provide the endpoints of the confidence interval.  The percentile method is a 
relatively simple calculation and has the advantage of not requiring an estimate of the standard error.  Unfortunately, 
what this method provides in simplicity and appeal also results in a tendency to be less successful for non-normal 
distributions. 
 
The MACRO D2BAND currently computes three different band-widths (i.e., 80%, 95%, 99%) using the 
aforementioned methods of estimation (i.e., the interval inversion approach for D2, and the bootstrapping method for 
the two modified multivariate effect sizes). 

 
The remainder of the paper will describe and present the MACRO D2BAND. 
 
MACRO D2BAND 
The macro D2BAND is called from SAS after the data have been read into a SAS data set. Inputs to the macro 
include the name of the SAS data set that contains the data to be analyzed (DSN), the name of the independent 
variable (Group) and the names of the dependent variables (DVs). 
 
%macro D2BAND(DSN,Group,DVs); 
 
proc iml; 
 
* +------------------------------------------------------------------------------------------------------------------+ 

Subroutine MAHL computes Mahalanobis Distance and the Jacknife Distance. 

Inputs are data matrices for two groups (G1 and G2). 

Outputs are D2 and D2_J 
  +------------------------------------------------------------------------------------------------------------------+; 
 
START MAHL(G1,G2,D2,D2_J); 
 p=ncol(G1);n1=nrow(G1);n2=nrow(G2); 
 AG1 = J(n1,1,1)||G1; 
 AG2 = J(n2,1,1)||G2; 
 SSCP1 = AG1` * AG1; 
 SSCP2 = AG2` * AG2; 
 W1 = SSCP1[2:P+1,2:P+1]; 
  do r = 1 to p; 
   do c = 1 to p; 
    if r=c then W1[r,c]=W1[r,c]- (SSCP1[1,c+1]##2/SSCP1[1,1]); 
    if r^=c then W1[r,c]=W1[r,c]- (1/SSCP1[1,1]#SSCP1[1,c+1]# SSCP1[1,r+1]); 
   end; 
  end; 
 
  W2 = SSCP2[2:P+1,2:P+1]; 
  do r = 1 to p; 
   do c = 1 to p; 
    if r = c then W2[r,c] = W2[r,c] - (SSCP2[1,c+1]##2 / SSCP2[1,1]); 
    if r^=c then W2[r,c]=W2[r,c]-(1/SSCP2[1,1]#SSCP2[1,c+1]# SSCP2[1,r+1]); 
   end; 
  end; 
 
  W = W1 + W2; 
  cent1 = J(1,n1,1) * G1 * (1/n1); 
  cent2 = J(1,n2,1) * G2 * (1/n2); 
  Cov_matrix=(1/(n1+n2-2))*W; 
  D2=(cent1-cent2)*INV(Cov_matrix) *(cent1-cent2)`; 
 
* +----------------------------------------------------------------------------------------------------+ 
   Remove each observation from G1 and recompute Mahalanobis Distance 
  +----------------------------------------------------------------------------------------------------+; 
D2_1 = 0; 
 do i = 1 to n1; 
  one_obs = AG1[i,]; 

  one_sscp = one_obs` * one_obs; 
  SSCP1_i = SSCP1 - one_sscp; 
  W1_i = SSCP1_i[2:P+1,2:P+1]; 
  do r = 1 to p; 
   do c = 1 to p; 
    if r=c then W1_i[r,c]=W1_i[r,c]-(SSCP1_i[1,c+1]##2/SSCP1_i[1,1]); 
    if r^=c then W1_i[r,c]=W1_i[r,c]-(1/SSCP1_i[1,1]#SSCP1_i[1,c+1]# SSCP1_i[1,r+1]); 
    end; 
  end; 
  W_i = W1_i + W2; 
 
 if i=1 then G1_i=G1[2:n1,]; 
 if (i>1 & i<n1) then G1_i=G1[1:i-1,]// G1[i+1:n1,]; 
 if i=n1 then G1_i=G1[1:n1-1,]; 
 cent1 = J(1,n1-1,1) * G1_i * (1/(n1-1)); 
 cent2 = J(1,n2,1) * G2 * (1/n2); 
 Cov_matrix = (1/(n1 + n2 - 3)) * W_i; 
 D2_1=D2_1+((cent1-cent2)*INV(Cov_matrix)* (cent1-cent2)`); 
end; 
 
* +--------------------------------------------------------------------------------------------------------+ 
   Remove each observation from G2 and recompute Mahalanobis Distance 
  +--------------------------------------------------------------------------------------------------------+; 
D2_2 = 0; 
 do i = 1 to n2; 
  one_obs = AG2[i,]; 
  one_sscp = one_obs` * one_obs; 
  SSCP2_i = SSCP2 - one_sscp; 
  W2_i = SSCP2_i[2:P+1,2:P+1]; 
  do r = 1 to p; 
   do c = 1 to p; 
    if r=c then W2_i[r,c]=W2_i[r,c]-(SSCP2_i[1,c+1]##2/SSCP2_i[1,1]); 
    if r^=c then W2_i[r,c]=W2_i[r,c]-(1/SSCP2_i[1,1]#SSCP2_i[1,c+1]# SSCP2_i[1,r+1]); 
   end; 
  end; 
  W_i = W1 + W2_i; 
  if i=1 then G2_i=G2[2:n2,]; 
  if (i>1 & i<n2) then G2_i=G2[1:i-1,]// G2[i+1:n2,]; 
  if i = n2 then G2_i = G2[1:n2-1,]; 
  cent1 = J(1,n1,1) * G1 * (1/n1); 
  cent2 = J(1,n2-1,1) * G2_i * (1/(n2-1)); 
  Cov_matrix = (1/(n1 + n2 - 3)) * W_i; 
  D2_2=D2_2+((cent1-cent2)*INV(Cov_matrix)* (cent1-cent2)`); 
 end; 
 D_bar = (D2_1 + D2_2)/(n1 + n2); 
 D2_J = (n1+n2)#D2-(n1+n2-1)#D_bar; 
FINISH; 
 
* +-----------------------------------------------------------------------------------------------------------------------------------+ 

Subroutine BOOTSTRP draws a random sample with replacement from a data matrix. 

Inputs are the original data matrix (RWDATA) and desired size of bootstrap sample (BOOTN). 

Output is the random sample (B_X) 
  +------------------------------------------------------------------------------------------------------------------------------------+; 
 
START BOOTSTRP (RWDATA, BOOTN, B_X); 
 ORIG_X = RWDATA; 
 BIG_N = NROW(ORIG_X); 
 DO i = 1 to BOOTN; 
  ranrow=round((ranuni(0)*BIG_N)+0.4999); 
  if ranrow = 0 then ranrow = 1; 
  if i = 1 then do; 
   B_X = ORIG_X[ranrow,]; 
  end; 
  if i>1 then do; 
   B_X = B_X//ORIG_X[ranrow,]; 
  end; 
 end; 
 free big_n i ranrow; 
FINISH; 

* +---------------------------------------------------------------------------------------------------------------------+ 
Subroutine BOOTP evaluates percentiles from a vector of bootstrapped statistics. 

Input is the vector of bootstrapped statistics (md_vec).  

Outputs are values of the statistic at defined percentiles. 
  +---------------------------------------------------------------------------------------------------------------------+; 
 
 
START BOOTP(md_vec,lo99,hi99,lo95,hi95, lo80,hi80); 
 Nboots = nrow(md_vec); 
 
 r_vec = rank(md_vec); 
 pct025 = round(nboots#.025); 
 pct975 = round(nboots#.975); 
 pct005 = round(nboots#.005); 
 pct995 = round(nboots#.995); 
 pct100 = round(nboots#.100); 
 pct900 = round(nboots#.900); 
 do bb= 1 to nboots; 
  if r_vec[bb]=pct025 then lo95=MD_vec[bb]; 
  if r_vec[bb]=pct975 then hi95=MD_vec[bb]; 
  if r_vec[bb]=pct005 then lo99=MD_vec[bb]; 
  if r_vec[bb]=pct995 then hi99=MD_vec[bb]; 
  if r_vec[bb]=pct100 then lo80=MD_vec[bb]; 
  if r_vec[bb]=pct900 then hi80=MD_vec[bb]; 
 end; 
FINISH; 
 
* +---------------------------------------------------------------------------------------------------------------------------------------------------------------------------+ 

Subroutine MAHL_D_PCTL computes percentiles from the sampling distribution of Mahalanobis Distance using the interval 
inversion method. 
Inputs are the sample value of D2 (smpl_MD), sample sizes in the two groups (n1 and n2), the number of variables in each 
group (p), and the desired percentile from the sampling distribution (pctl). 
Output is population value of D2 that provides smpl_MD at the pctl percentile. 

Subroutine returns the value 999 if the percentile cannot be obtained numerically 
 +---------------------------------------------------------------------------------------------------------------------------------------------------------------------------+; 
 
START MAHL_D_PCTL(smpl_MD,n1,n2,p,pctl, mahl_d_t); 
  if smpl_MD<0 then smpl_MD=0; 
  d2_tilde=(((n1*n2)*(n1+n2-p-1))/ ((n1+n2)*(n1+n2-2)*p))*smpl_MD; 
  df1 = p; 
  df2 = n1+n2-p-1; 
     
  * Step 1: Find value of D2 that is a little too high; 
  OK = 0; 
  oops = 0; 
  mahl_d_t = 0; * start the loop with D2 = 0; 
  do until (OK = 1); 
   nc = mahl_d_t # ((n1#n2)/(n1+n2)); 
   if nc < 5500 then do; 
    cumprob = PROBF(d2_tilde,df1,df2,nc); 
    if cumprob<pctl then OK = 1; 
    if cumprob>pctl then mahl_d_t=mahl_d_t+1; 
   end; 
   if nc >= 5500 then do; 
    mahl_d_t = 999; 
    OK = 1; 
   end; 
  end; 
  high = mahl_d_t; 
 
  * Step 2: Establish value of D2 that is a little too low (D2 = 0); 
  low = 0; 
 
  * Step 3: Successively halve the interval between low and high to obtain final value of percentile; 
 
  change = 1; 
  loop = 0; 
  small = .00000000001; 
  do until (change<small); 
   half = (high + low)/2; 

   nc = half # ((n1#n2)/(n1+n2)); 
   if nc < 5500 then do; 
    cumprob = PROBF(d2_tilde,df1,df2,nc); 
    if cumprob < pctl then high =  half;        
    if cumprob > pctl then low  = half; 
    change = abs(high - low); 
    loop = loop + 1; 
    if loop > 1500 then small = .000000001; 
    if loop > 3000 then small = .01; 
    mahl_d_t = (high + low)/2; 
  end; 
  if nc >= 5500 then do; 
   mahl_d_t = 999; 
   change = 0; 
  end; 
 end; 
 FINISH; 
 
* +--------------------+ 
   Main Program 
  +--------------------+; 
 
* +--------------------------------------------------+ 
   Read data from regular SAS into IML 
  +--------------------------------------------------+; 
 use &DSN; 
 read all var{&DVs} where (&Group=1) into G1;  
 read all var{&DVs} where (&Group=2) into G2;  
 n1 = nrow(G1); 
 n2 = nrow(G2); 
 p = ncol(G1); 
 
* +---------------------------------------------------------------------------+ 

Compute D2, Jacknife D2, and Rao adjusted D2 
  +---------------------------------------------------------------------------+; 
run mahl(G1, G2, MD, MD_J); 
D2_u1=((n1 + n2 - p -3)/(n1 + n2 -2))*MD; 
D2_u2= p*(1/n1 + 1/n2); 
D2_u= D2_u1 - D2_u2; 
 
* +------------------------------------------------------------------------------------------+ 

Compute endpoints of interval inversion confidence bands for D2 
  +------------------------------------------------------------------------------------------+; 
 run mahl_d_PCTL(MD,n1,n2,p,0.005,hi_I_99); 
 run mahl_d_PCTL(MD,n1,n2,p,0.995,lo_I_99); 
 run mahl_d_PCTL(MD,n1,n2,p,0.025,hi_I_95); 
 run mahl_d_PCTL(MD,n1,n2,p,0.975,lo_I_95); 
 run mahl_d_PCTL(MD,n1,n2,p,0.100,hi_I_80); 
 run mahl_d_PCTL(MD,n1,n2,p,0.900,lo_I_80); 
 
* +----------------------------------------------------------------------------------------------------------------------+ 

Generate 5000 Bootstrap Samples and place bootstrap statistics in matrix boot_mtx 
  +----------------------------------------------------------------------------------------------------------------------+; 
do bb = 1 to 5000; 
 run bootstrp (G1, n1, b1); 
 run bootstrp (G2, n2, b2); 
 run mahl(b1, b2, D2b, MDJ_b); 
 D2_u1b=((n1 + n2 - p -3)/(n1 + n2 -2))*D2b; 
 D2_ub= D2_u1b - D2_u2; 
 if bb=1 then BOOT_MTX = MDJ_b||D2_ub; 
 if bb>1 then BOOT_MTX=BOOT_MTX// (MDJ_b||D2_ub); 
 free b1 b2 MD_b MDJ_b D2_ub; 
end;  
 
* +----------------------------------------------------------------------------------------------------+ 

Compute endpoints of bootstrap CI for Jacknife and Rao adjusted D2 
  +----------------------------------------------------------------------------------------------------+; 
 test_vec = BOOT_MTX[,1];   
 run BOOTP(test_vec,l9j,h9j,l5j,h5j,l0j,h0j); 
 test_vec = BOOT_MTX[,2]; 

 run BOOTP(test_vec,l9r,h9r,l5r,h5r,l0r,h0r); 
 
* +-----------------------------------------+ 
              Print output results 
  +-----------------------------------------+; 
file print; 
put @1 'Mahalanobis Distance Estimates & Intervals' // 
@1 'Mahalanobis Distance'/ 
@1 '--------------------------------------------------'/ 
@1 'N of Variables' @47 p 3. // 
@1 'Sample Sizes' @41 n1 4. @45 ',' @46 n2 4. // 
@1 '--------------------------------------------------'/ 
@1 '       Interval Inversion Confidence Bands' / 
@1 '--------------------------------------------------'/// 
@1 'Sample Value of D2' @42 MD 8.5 // 
@1 'Confidence Bands' @25 'Lower Limit'  
@40 'Upper Limit' / 
@1 '----------------' @25 '-----------'  
@40 '-----------' / 
@7 '80%' @27 lo_I_80 8.5 @42 hi_I_80 8.5 / 
@7 '95%' @27 lo_I_95 8.5 @42 hi_I_95 8.5 / 
@7 '99%' @27 lo_I_99 8.5 @42 hi_I_99 8.5 /// 
@1 '--------------------------------------------------'/ 
@1 '            Bootstrap Confidence Bands' / 
@1 '--------------------------------------------------'/// 
@1 'Sample Value of Jacknife D2' @42 MD_J 8.5 // 
@1 'Confidence Bands' @25 'Lower Limit'  
@40 'Upper Limit' / 
@1 '----------------' @25 '-----------'  
@40 '-----------' / 
@7 '80%' @27 l0j 8.5 @42 h0j 8.5 / 
@7 '95%' @27 l5j 8.5 @42 h5j 8.5 / 
@7 '99%' @27 l9j 8.5 @42 h9j 8.5 // 
@1 '--------------------------------------------------'/// 
@1 'Sample Value of Rao Adjusted D2' @42 D2_U 8.5 // 
@1 'Confidence Bands' @25 'Lower Limit'  
@40 'Upper Limit' / 
@1 '----------------' @25 '-----------'  
@40 '-----------' / 
@7 '80%' @27 l0r 8.5 @42 h0r 8.5 / 
@7 '95%' @27 l5r 8.5 @42 h5r 8.5 / 
@7 '99%' @27 l9r 8.5 @42 h9r 8.5 // 
@1 '--------------------------------------------------'; 
quit; 
%mend D2BAND; 
 
EXAMPLE OF MACRO INVOCATION 
An example of the macro’s use is provided by a study of gender differences on four psychological traits. A sample of 
32 undergraduate male and 32 undergraduate female psychology students completed measures of each trait. The 
data were analyzed by testing for differences in population mean vectors (using Hotelling’s T2 ) and by estimating the 
magnitude of the multivariate difference using Mahalanobis Distance.  As the following code illustrates, the data were 
read into a SAS data set (GEN_DIFFS), and the macro D2BAND was called using a run statement. The arguments to 
the macro included the name of the SAS data set, the name of the grouping variable (GENDER), and the name of the 
dependent variables for the analysis (T1 – T4).  The results of this analysis are presented in Table 1. 
 
data gen_diffs; 
 input gender 1 t1 3-4 t2 6-7 t3 9-10 t4 12-13; 
cards; 
1 15 17 24 14 
1 17 15 32 26 
. 
. 
2 14 16 22 26 
2 17 17 22 28 
; 
%D2BAND(gen_diffs,gender,t1 t2 t3 t4); run; 
 
OUTPUT FROM MACRO D2BAND 
Table 1 provides an example of the output produced by the macro D2BAND.   

Table 1 
 
Mahalanobis Distance Estimates & Intervals 
 
Mahalanobis Distance 
------------------------------------------------------------------ 
N of Variables                              4  
 
Sample Sizes                              32,  32 
------------------------------------------------------------------ 
   Interval Inversion Confidence Bands 
------------------------------------------------------------------ 
 
Sample Value of D2                               6.10009 
 
Confidence Bands      Lower Limit    Upper Limit 
-------------------------      ----------------   ---------------- 
      80%                      3.71199          7.80135 
      95%                        2.88844          9.15019 
      99%                        2.23190        10.47390 
------------------------------------------------------------------ 
   Bootstrap Confidence Bands 
------------------------------------------------------------------ 
 
Sample Value of Jacknife D2                 5.13510 
 
Confidence Bands      Lower Limit    Upper Limit 
-------------------------      ----------------  ----------------- 
     80%                         3.98709          8.68028 
     95%                         3.27355        10.58943 
     99%                         2.74402        12.65897 
  ------------------------------------------------------------------ 
 
Sample Value of Rao Adjusted D2         5.35815 
 
Confidence Bands      Lower Limit    Upper Limit 
-------------------------      ----------------    ---------------- 
      80%                        4.17527          8.98404 
      95%                        3.41742        10.94231 
      99%                        2.81740        13.14760 
-------------------------------------------------------------------- 
 
RESULTS 
For these data, the sample value of D2 is 6.10009, a very large multivariate difference between the males and 
females. Because D2 is known to be positively biased in small samples, better point estimates of the multivariate 
difference are provided by the Jacknife D2 (5.13510) or Rao’s adjusted D2 (5.35815).  The interval estimates of the 
population value of D2 reflect the large amount of uncertainty about this parameter, because of the small sample of 
research participants. For example, the 95% confidence interval constructed using interval inversion ranges from 
2.88844 to 9.15019. Somewhat wider confidence intervals are obtained from the bootstrap estimates. 
 
EMPIRICAL STUDIES OF POINT ESTIMATES AND CONFIDENCE INTERVALS 
As stated earlier, in a recent Monte Carlo study we investigated the accuracy and precision of three point estimates 
and confidence bands for two-group multivariate effect sizes under a variety of data conditions, including nonnormality 
of population distributions and covariance matrix heterogeneity.  The results of our study were initially analyzed by 
examining confidence band coverage probabilities and confidence band-widths.  With respect to confidence band 
coverage estimates for Mahalanobis Distance (D2), the interval inversion approach emerged as the superior method, 
outperforming the percentile approach to the bootstrap.  When examining the performance of the two computational 
methods employing the Jackknife estimate of D2, a different picture emerged.  In this case, the interval inversion 
method exhibited a rapid deterioration, while the percentile bootstrap method improved dramatically.  Similarly, when 
we compared the two confidence band estimation methods for the Rao adjusted D2, the superior performance of the 
percentile bootstrap was evident, while the interval approach reported probability coverage levels below 50%.  

Confidence band-width for the three multivariate effect size indices was computed by employing the interval inversion 
approach for D2 and the percentile bootstrap methods for the other two estimates.  An examination of the three 
estimates of D2 , revealed that the Jackknife and Rao estimates of D2 evidenced more variability across the conditions 
than the interval approach for D2. 
 
Although the focus of this study was on interval estimates of effect size, the bias in each of the indices as point 
estimates of multivariate effect size was also examined.  The Jackknife estimate of D2 appeared to exhibit less 
statistical bias than the other two methods of computing D2.  The original Mahalanobis D2 was observed to evidence 
considerable bias on average and considerable variability in bias across the conditions examined.  The Jackknife and 
Rao estimates appeared to fare better, evidencing little bias and appearing fairly consistent across conditions.  
However, the box and whisker plots indicate that the Rao estimate was slightly more biased across more conditions 
than the Jackknife estimate. 
 
CONCLUSION 
Naturally, this macro can also be modified to other contexts depending on the specifications of the user.  For 
example, the macro currently calculates 80%, 95%, and 99% confidence intervals.  Simple modifications of the macro 
will produce 90% confidence intervals or one-sided confidence intervals. Additionally, the macro currently estimates 
bootstrap confidence intervals based on 5000 bootstrap samples (the value hard coded in the bootstrapping DO 
loop). If greater precision in the estimation of the endpoints of these bootstrap intervals is desired, this value can be 
increased. The precision of the estimation of confidence interval endpoints is proportional to the number of bootstrap 
samples drawn.   

